#/*-----------------------------------------------------------------*/
#  build the documentation
#
#  Copyright, 2023-2025, CNRS
#   email of the author : Mickael.Gastineau@obspm.fr
#
#/*-----------------------------------------------------------------*/
#/*-----------------------------------------------------------------*/
#/* License  of this file :
#  This file is "triple-licensed", you have to choose one  of the three licenses 
#  below to apply on this file.
#  
#     CeCILL-C
#     	The CeCILL-C license is close to the GNU LGPL.
#     	( http://www.cecill.info/licences/Licence_CeCILL-C_V1-en.html )
#   
#    or CeCILL-B
#        The CeCILL-B license is close to the BSD.
#        (http://www.cecill.info/licences/Licence_CeCILL-B_V1-en.txt)
#  
#    or CeCILL v2.1
#        The CeCILL license is compatible with the GNU GPL.
#        ( http://www.cecill.info/licences/Licence_CeCILL_V2.1-en.html )
#  
# 
# This library is governed by the CeCILL-C, CeCILL-B or the CeCILL license under 
# French law and abiding by the rules of distribution of free software.  
# You can  use, modify and/ or redistribute the software under the terms 
# of the CeCILL-C,CeCILL-B or CeCILL license as circulated by CEA, CNRS and INRIA  
# at the following URL "http://www.cecill.info". 
#
# As a counterpart to the access to the source code and  rights to copy,
# modify and redistribute granted by the license, users are provided only
# with a limited warranty  and the software's author,  the holder of the
# economic rights,  and the successive licensors  have only  limited
# liability. 
#
# In this respect, the user's attention is drawn to the risks associated
# with loading,  using,  modifying and/or developing or reproducing the
# software by the user in light of its specific status of free software,
# that may mean  that it is complicated to manipulate,  and  that  also
# therefore means  that it is reserved for developers  and  experienced
# professionals having in-depth computer knowledge. Users are therefore
# encouraged to load and test the software's suitability as regards their
# requirements in conditions enabling the security of their systems and/or 
# data to be ensured and,  more generally, to use and operate it in the 
# same conditions as regards security. 
#
# The fact that you are presently reading this means that you have had
# knowledge of the CeCILL-C,CeCILL-B or CeCILL license and that you accept its terms.
# */
# /*-----------------------------------------------------------------*/
set(CMAKE_MODULE_PATH "${CMAKE_SOURCE_DIR}/toolchain")

configure_file(source/confcommon.py.in source/confcommon.py)

find_package(SphinxDoc)
if (ENABLE_PDF)
    find_package(LATEX COMPONENTS PDFLATEX)
endif()

if(NOT TARGET doc)
    add_custom_target(doc)
endif()

# function to build the documentation for one language
function(builddoc languagesource languagecommon languagename)
    configure_file(${languagesource}/conf.py.in ${languagesource}/conf.py)

        # build the html documentation for one language
    if (ENABLE_DOC)
    add_custom_target(doc_${languagename}_html
        ${SPHINXDOC_EXECUTABLE} -b html -d "${CMAKE_CURRENT_BINARY_DIR}/sphinxdoctrees_html_${languagename}" -c "${CMAKE_CURRENT_BINARY_DIR}/${languagesource}/" ${languagecommon}  "${CMAKE_CURRENT_BINARY_DIR}/html/${languagename}"
        WORKING_DIRECTORY "${CMAKE_CURRENT_SOURCE_DIR}"
        COMMENT "build the HTML documentation for the language ${languagename}")
    add_dependencies(doc doc_${languagename}_html)  
    
    install(DIRECTORY "${CMAKE_CURRENT_BINARY_DIR}/html/${languagename}" DESTINATION "${CMAKE_INSTALL_DATADIR}/doc/calceph/html/" )

    endif(ENABLE_DOC)
    
    # build the pdf documentation for one language
    if (ENABLE_PDF)
        if(LATEX_PDFLATEX_FOUND)
            set(latexdir "${CMAKE_CURRENT_BINARY_DIR}/latex/${languagename}")
            add_custom_target(doc_${languagename}_pdf
                ${SPHINXDOC_EXECUTABLE} -b latex -d "${CMAKE_CURRENT_BINARY_DIR}/sphinxdoctrees_pdf_${languagename}" -c "${CMAKE_CURRENT_BINARY_DIR}/${languagesource}/" ${languagecommon}  ${latexdir}
                WORKING_DIRECTORY "${CMAKE_CURRENT_SOURCE_DIR}"
                COMMENT "build the pdf documentation for the language ${languagename}")

                # get all xrefs using three compilation
            add_custom_command(TARGET doc_${languagename}_pdf POST_BUILD
                COMMAND "${PDFLATEX_COMPILER}" "calceph.tex"
                COMMAND "${PDFLATEX_COMPILER}" "calceph.tex"
                COMMAND "${PDFLATEX_COMPILER}" "calceph.tex"
                COMMAND ${CMAKE_COMMAND} -E copy "${latexdir}/calceph.pdf"  "${CMAKE_CURRENT_BINARY_DIR}/calceph_${languagename}.pdf"
                WORKING_DIRECTORY "${latexdir}")

                add_dependencies(doc doc_${languagename}_pdf)  

                install(FILES "${CMAKE_CURRENT_BINARY_DIR}/calceph_${languagename}.pdf" DESTINATION "${CMAKE_INSTALL_DATADIR}/doc/calceph/")

        endif() 
    endif()
endfunction()



builddoc(source_c source c )
builddoc(source_f2003 source fortran2003 )
builddoc(source_f9x source fortran9x )
builddoc(source_python source python )
builddoc(source_mex source mex )

